[bookmark: _GoBack]$title Cadena de suministro inversa llantas usadas Ibagué
Sets
    i       'Conjunto de ubicaciones de los generadores de llantas usadas'      /i1*i18/
    j       'Conjunto de alternativas de ubicación para los centros de acopio'  /Salado,Picalena,ElPapayo,Miramar/
    k       'Conjunto de ubicaciones de plantas de recuperaciónn'               /Bogota,Ibague,Medellin/
    t       'Conjunto de alternativas tecnológicas a implementar'               /t1*t8/
    a       'Periodos de tiempo'                                                /a1*a8/;
Parameters
    CA(t,j)      'Costo de apertura del centro de acopio en la ubicación j con la tecnología t'
    CVP(t,a)     'Costo unitario de producto fabricado con la tecnología t en el año a'
    CTL(a)       'Costos de transporte de llantas usadas en el año a'
    CTP(a)       'Costos de transporte de GRC y productos terminados en el año a'
    DG(i,j)      'Distancia entre el generador i y el centro de acopio j'
    DP(j,k)      'Distancia entre el centro de acopio j y la panta de recuperación k'
    DEMG(k,a)    'Demanda de GRC de la planta de recuperación k en el año a'
    DEML(k,a)    'Demanda de LFU recauchutables de la planta de recuperación k en el año a'
    PV(t,a)      'Precio de mercado del producto fabricado con la tecnología t en el año a'
    PVR(a)       'Precio de mercado de llantas recauchutables en el año a'
    PVA(a)       'Precio de mercado del acero en el año a'
    PC(a)        'Precio de compra de llantas usadas recolectadas en el año a'
    CM(j,a)      'Costo unitario de manejo en el centro de acopio j en el año a'
    LFU(i)       'Generación de llantas usadas en la ubicación i'
    ECOA(t)      'Eco indicador por tonelada producida de GRC utilizando la tecnología t'
    ECOB(t)      'Eco indicador por unidad producida utilizando la tecnología t'
    ECOD(k)      'Eco indicador por tonelada de GRC enviada a la planta de recuperación k'
    FC(t)        'Factor de conversión utilizando la tecnología t'
    SOCGE(t)     'Puntaje del subcriterio generación de empleo para alternativa tecnológica t'
    SOCSL(t)     'Puntaje del subcriterio seguridad laboral para alternativa tecnológica t'
    SOCDL(j)     'Puntaje del subcriterio desarrollo local para la ubicación j'
    CLFU(a)      'Factor de crecimiento de llantas usadas en el año a'
    INT(a)       'Factor de conversión del dinero en el tiempo en el año a'
    MR(a)        'Meta de recolección de residuos en el año a'
    CAPG(t,j)    'Capacidad de producción de GRC del centro de acopio en la ubicación j con la tecnología t'
    CAPL(j)      'Capacidad de recepción de LFU del centro de acopio en la ubicación j'
    CAPP(t,j)    'Capacidad del centro de acopio en la ubicación j utilizando la tecnología t'
    Peso         'Peso promedio de una llanta usada en toneladas'
    /0.01485/
    ECOTA        'Eco indicador asociado al transporte de LFU'
    /0.2/
    ECOTB        'Eco indicador asociado al transporte de LFU recauchatables'
    /0.07/
    ECOTC        'Eco indicador asociado al transporte de GRC y producto terminado'
    /0.05/
    ECOC         'Eco indicador por LFU enviada a proceso de recauchatado k'
    /-3.95/
    FLR          'Proporción de llantas usadas susceptibles de recauchutado'
    /0.28/
    AC           'Composición de acero de las llantas usadas'
    /0.18/
    FCT          'Factor de conversión para el proceso de trituración caucho'
    /96.15/
    WGE          'Ponderación del subcriterio social generación de empleo'
    /0.4419/
    WSL          'Ponderación del subcriterio social seguridad laboral'
    /0.1198/
    WDL          'Ponderación del subcriterio social desarrollo local'
    /0.4383/
    CCA          'Número máximo de centros de acopio a abrir en la ciudad'
    /1/
    M            'Número muy grande'
    /9999999999999999/;
*Lectura de datos de Excel
$onEcho > datos.txt
par=DG      rng=Tablas!A1:E23       cDim=1 rDim=1
par=PV      rng=Tablas!H1:P9        cDim=1 rDim=1
par=CVP     rng=Tablas!H12:P20      cDim=1 rDim=1
par=CM      rng=Tablas!H23:P27      cDim=1 rDim=1
par=CA      rng=Tablas!A26:E34      cDim=1 rDim=1
par=DEMG    rng=Tablas!H30:P33      cDim=1 rDim=1
par=DEML    rng=Tablas!H36:P39      cDim=1 rDim=1
par=CAPG    rng=Tablas!A37:E45      cDim=1 rDim=1
par=CAPP    rng=Tablas!A48:E56      cDim=1 rDim=1
par=DP      rng=Tablas!A59:D63      cDim=1 rDim=1
par=CTL     rng=Vectores!A1:B9      rDim=1
par=CTP     rng=Vectores!E1:F9      rDim=1
par=PVA     rng=Vectores!I1:J9      rDim=1
par=PC      rng=Vectores!M1:N9      rDim=1
par=PVR     rng=Vectores!Q1:R9      rDim=1
par=INT     rng=Vectores!U1:V9      rDim=1
par=MR      rng=Vectores!Y1:Z9      rDim=1
par=CLFU    rng=Vectores!AC1:AD9    rDim=1
par=ECOA    rng=Vectores!A12:B20    rDim=1
par=ECOB    rng=Vectores!E12:F20    rDim=1
par=FC      rng=Vectores!I12:J20    rDim=1
par=SOCGE   rng=Vectores!M12:N20    rDim=1
par=SOCSL   rng=Vectores!Q12:R20    rDim=1
par=ECOD    rng=Vectores!A23:B26    rDim=1
par=SOCDL   rng=Vectores!A29:B33    rDim=1
par=CAPL    rng=Vectores!E29:F33    rDim=1
par=LFU     rng=Vectores!A36:B58    rDim=1
$offEcho
$call GDXXRW anexo_3.xlsx @datos.txt trace=0
$gdxIn anexo_3.gdx
$load DG PV CVP CM CA DEMG DEML CAPG CAPP DP CTL CTP PVA PC PVR INT MR CLFU ECOA ECOB FC SOCGE SOCSL ECOD SOCDL CAPL LFU
$gdxIn
Display DG,PV,CVP,CM,CA,DEMG,DEML,CAPG,CAPP,DP,CTL,CTP,PVA,PC,PVR,INT,MR,CLFU,ECOA,ECOB,FC,SOCGE,SOCSL,ECOD,SOCDL,CAPL,LFU;
$Call sql2gms C="Provider=Microsoft.Jet.OLEDB.4.0;Data Source=anexo.mdb" Q="SELECT j, t, a, CF FROM Tabla1" O=CF.inc
$Call sql2gms C="Provider=Microsoft.Jet.OLEDB.4.0;Data Source=anexo.mdb" Q="SELECT k, t, a, CVT FROM Tabla2" O=CVT.inc
$Call sql2gms C="Provider=Microsoft.Jet.OLEDB.4.0;Data Source=anexo.mdb" Q="SELECT k, t, a, PVG FROM Tabla3" O=PVG.inc
parameters
    CF(j,t,a)    'Costo fijo de operación del centro de acopio en la ubicación j con la tecnología t en el año a'/
$include CF.inc/
    CVT(k,t,a)   'Costo por tonelada triturada utilizando la tecnología t con destino a la planta de recuperación k en el año a'/
$include CVT.inc/
    PVG(k,t,a)   'Precio de mercado de GRC fabricado con la tecnología t en la planta de recuperación k en el año a'/
$include PVG.inc/;
Display CF, CVT, PVG;
Variable
    X(j,t)       'Variable binaria 1 para el centro de acopio establecido en la ubicación j implementando la tecnología t, de otro modo es 0'
    XL(i,j,a)    'Flujo de LFU desde el generador i hasta el centro de acopio j en el año a'
    XLR(j,k,a)   'Flujo de LFU recauchutables desde el centro de acopio j hasta la planta de recuperación k en el año a'
    XGR(j,k,t,a) 'Flujo de granulo de caucho desde el centro de acopio j hasta la planta de recuperación k utilizando la tecnología t en el año a'
    XPG(j,t,a)   'Producción de GRC en el centro de acopio j con la tecnología t en el año a'
    XPT(j,t,a)   'Fabricación de producto terminado en el centro de acopio j con la tecnología t en el año a'
    VPN          'Valor de la función objetivo considerando factores economicos'
    P(j,a)       'perdida'
    U(a)         'Utilidad en el año a'
    ING(a)       'Ingresos en el año a'
    COS(a)       'Costos totales en el año a'
    ECO99        'Valor de la función objetivo considerando factores Ambientales'
    Social       'Valor de la función objetivo considerando factores sociales'
    Objetivo     'Función que incorpora los tres objetivos';
Binary   Variable X;
Positive Variable XL, XLR, XGR, XPG, XPT;
parameter     C(a)       'aumento en la capacidad'
/ a1     1	a2     1         a3     1         a4     1         a5     1.2         a6     1.2         a7     1.2         a8     1.2      /;
Equation
    Ut(a)        'Utilidad en el año a'
    In(a)        'Ingreso en el año a'
    Ct(a)        'Costo total en el año a'
    Gen(i,a)     'Restricción de generación de llantas usadas'
    Dema(k,a)    'Restricción de demanda de granulo de caucho'
    Demb(k,a)    'Restricción de demanda de llantas usadas recauchutables'
    Capa(j,a)   'Restricción de capacidad de gestión de llantas usadas'
    Capb(j,t,a)  'Restricción de capacidad de producción de GRC'
    Capc(j,t,a)  'Restricción de capacidad de fabricación de producto terminado'
    Flua(j,a)    'Restricción de flujo de llantas usadas'
    Flub(j,t,a)  'Restricción de flujo de granulo de caucho'
    Rec(j,a)     'Porción de llantas usadas susceptibles de recauchutado'
    Loc          'Apertura de un solo centro de acopio'
    Vp           'VPN mayor o igual a 0'
    Meta(i,a)    'Meta de recolección'
    GranMa(j)    'No envios a centros no abiertos'
    GranMb(j)    'No envios a centros no abiertos'
    GranMc(j,t)  'No envios a centros no abiertos'
    GranMd(j,t)  'No envios a centros no abiertos'
    GranMe(j,t)  'No envios a centros no abiertos'
    Z1           'Maximizar VPN'
    Z2           'Minimizar Eco-indicador'
    Z3           'Maximizar componente social'
    Multiobj     'Función objetivo compuesta';
Ut(a)..         U(a) =e= ING(a)- Cos(a);
In(a)..         ING(a) =e= sum{j, sum(t, PV(t,a)*XPT(j,t,a)) + sum(k, PVR(a)*XLR(j,k,a)) + sum((k,t), 1000*PVG(k,t,a)*XGR(j,k,t,a)) + sum(t, (AC/(1-AC))*PVA(a)*XPG(j,t,a))};
Ct(a)..         Cos(a) =e= sum{j, sum(t, CF(j,t,a)*X(j,t)) + sum((k,t), 1000*CVT(k,t,a)*XGR(j,k,t,a)) + sum(t, 1000*CVT('Medellin',t,a)*(XPG(j,t,a)- sum(k, XGR(j,k,t,a)))) +
                sum(t, CVP(t,a)*XPT(j,t,a)) + sum(i, CTL(a)*DG(i,j)*XL(i,j,a)) + sum(k, CTL(a)*DP(j,k)*XLR(j,k,a))+ sum((k,t), CTP(a)*DP(j,k)*XGR(j,k,t,a)) + sum(i, (PC(a)+CM(j,a))*XL(i,j,a))};
Gen(i,a)..      sum(j, XL(i,j,a)) =l= CLFU(a)*LFU(i);
Dema(k,a)..     sum((j,t), XGR(j,k,t,a)) =l= DEMG(k,a);
Demb(k,a)..     sum(j, XLR(j,k,a)) =l= DEML(k,a);
Capa(j,a)..     sum(i, XL(i,j,a)) =l= CAPL(j)*C(a);
Capb(j,t,a)..   XPG(j,t,a) =l= CAPG(t,j)*C(a);
Capc(j,t,a)..   XPT(j,t,a) =l= CAPP(t,j)*C(a);
Flua(j,a)..     sum(i, XL(i,j,a)) =e= sum(k, XLR(j,k,a))+ sum(t, FCT*XPG(j,t,a));
Flub(j,t,a)..   XPG(j,t,a) =e= sum(k, XGR(j,k,t,a))+ FC(t)*XPT(j,t,a);
Rec(j,a)..      sum(k, XLR(j,k,a)) =l= sum(i, FLR*XL(i,j,a));
Loc..           sum((j,t), X(j,t)) =e= CCA;
Vp..            VPN =g= 689069172;
Meta(i,a)..     sum(j, XL(i,j,a)) =g= LFU(i)*CLFU(a)*MR(a);
GranMa(j)..     sum((i,a), XL(i,j,a)) =l= sum(t, M*X(j,t));
GranMb(j)..     sum((k,a), XLR(j,k,a)) =l= sum(t, M*X(j,t));
GranMc(j,t)..   sum((k,a), XGR(j,k,t,a)) =l= M*X(j,t);
GranMd(j,t)..   sum(a, XPT(j,t,a)) =l= M*X(j,t);
GranMe(j,t)..   sum(a, XPG(j,t,a)) =l= M*X(j,t);
Z1..            VPN =e= -sum((j,t), 0*CA(t,j)*X(j,t))+ sum(a, INT(a)*U(a));
Z2..            ECO99 =e= sum{a, sum((i,j), Peso*ECOTA*DG(i,j)*XL(i,j,a))+ sum((t,j), ECOB(t)*XPT(j,t,a))
                + sum((j,k), Peso*ECOTB*DP(j,k)*XLR(j,k,a))+ sum((j,k,t), ECOD(k)*XGR(j,k,t,a))
                + sum((j,t), ECOA(t)*XPG(j,t,a))+ sum((j,k), Peso*ECOC*XLR(j,k,a))+ sum((j,k,t), Peso*ECOTC*DP(j,k)*XGR(j,k,t,a))};
Z3..            Social =e= WGE*sum((j,t), SOCGE(t)*X(j,t))+ WSL*sum((j,t), SOCSL(t)*X(j,t))+ WDL*sum((j,t), SOCDL(j)*X(j,t));
Multiobj..      Objetivo =e= (0.60/4185931390)*VPN + (0.27/(-55690))*ECO99 + (0.14/0.204359461859215)*Social;
Model modelo /All/;
Option mip = Cplex;
modelo.optCr = 0;
solve modelo maximizing VPN using mip;
solve modelo minimizing ECO99 using mip;
solve modelo maximizing Social using mip;
solve modelo maximizing Objetivo using mip;
Parameters
VenProdTerm(a) 'Venta de producto terminado'
VentaLlantas(a)
VentasGranulo(a)
VentaAcero(a)
CostoFijo(a)
CostoPGranulado(a)
CostoPTerminado(a)
CostoRecoleccion(a)
CostoDistriLlantas(a)
CostoDistriGranulo(a)
CostoCompraLlantas(a)
CostoManejo(a)
Recoleccion(a)
EcoRecoleccion
EcoDistribucionRe
EcoDistribucionGra
EcoGranulo
EcoProduccion
EcoRecuperacion
EcoReca
Ocupaciona(a,j)
Ocupacionb(a,t,j)
Ocupacionc(a,t,j)
TasaRecoleccion(a);
VenProdTerm(a) = sum((j,t), PV(t,a)*XPT.l(j,t,a));
VentaLlantas(a) = sum((j,k), PVR(a)*XLR.l(j,k,a));
VentasGranulo(a) = sum((j,k,t), 1000*PVG(k,t,a)*XGR.l(j,k,t,a));
VentaAcero(a) = sum((j,t), (AC/(1-AC))*PVA(a)*XPG.l(j,t,a));
CostoFijo(a) = sum((j,t), CF(j,t,a)*X.l(j,t));
CostoPGranulado(a)= sum((j,k,t), 1000*CVT(k,t,a)*XGR.l(j,k,t,a))+sum((j,t), 1000*CVT('Medellin',t,a)*(XPG.l(j,t,a)- sum(k, XGR.l(j,k,t,a))));
CostoPTerminado(a)= sum((j,t), CVP(t,a)*XPT.l(j,t,a));
CostoRecoleccion(a)= sum((i,j), CTL(a)*DG(i,j)*XL.l(i,j,a));
CostoDistriLlantas(a)= sum((j,k), CTL(a)*DP(j,k)*XLR.l(j,k,a));
CostoDistriGranulo(a)= sum((j,k,t), CTP(a)*DP(j,k)*XGR.l(j,k,t,a));
CostoCompraLlantas(a)= sum((j,i), PC(a)*XL.l(i,j,a));
CostoManejo(a)= sum((j,i), CM(j,a)*XL.l(i,j,a));
Recoleccion(a)= sum((i,j), XL.l(i,j,a));
EcoRecoleccion= sum((a,i,j), Peso*ECOTA*DG(i,j)*XL.l(i,j,a));
EcoDistribucionRe= sum((a,j,k), Peso*ECOTB*DP(j,k)*XLR.l(j,k,a));
EcoDistribucionGra= sum((a,j,k,t), Peso*ECOTC*DP(j,k)*XGR.l(j,k,t,a));
EcoGranulo= sum((a,j,t), ECOA(t)*XPG.l(j,t,a));
EcoProduccion= sum((a,t,j), ECOB(t)*XPT.l(j,t,a));
EcoRecuperacion=sum((a,j,k,t), ECOD(k)*XGR.l(j,k,t,a));
EcoReca= sum((a,j,k), Peso*ECOC*XLR.l(j,k,a));
Ocupaciona(a,j) = sum(i, XL.l(i,j,a))/(CAPL(j)*C(a));
Ocupacionb(a,t,j)= XPG.l(j,t,a)/(CAPG(t,j)*C(a));
Ocupacionc(a,t,j)= XPT.l(j,t,a)/(CAPP(t,j)*C(a));
TasaRecoleccion(a)= sum((i,j),XL.l(i,j,a))/sum(i,CLFU(a)*LFU(i));
Display X.l, XL.l, XLR.l, XGR.l, XPG.l, XPT.l, VPN.l, ECO99.l, Social.l,VenProdTerm,
VentaLlantas,VentasGranulo,VentaAcero,Recoleccion,CostoPGranulado, CostoFijo,CostoManejo,
CostoPTerminado,CostoRecoleccion,CostoDistriLlantas, CostoDistriGranulo,CostoCompraLlantas,
EcoRecoleccion,EcoDistribucionRe,EcoDistribucionGra,EcoGranulo,EcoProduccion,EcoRecuperacion,
EcoReca,Ocupaciona,Ocupacionb,Ocupacionc, TasaRecoleccion;
execute_unload "results.gdx" X.l, XL.l, XLR.l, XGR.l, XPG.l, XPT.l, VPN.l, ECO99.l, Social.l,VenProdTerm,
VentaLlantas,VentasGranulo,VentaAcero,Recoleccion,CostoPGranulado, CostoFijo,CostoManejo,
CostoPTerminado,CostoRecoleccion,CostoDistriLlantas, CostoDistriGranulo,CostoCompraLlantas,
EcoRecoleccion,EcoDistribucionRe,EcoDistribucionGra,EcoGranulo,EcoProduccion,EcoRecuperacion,
EcoReca,Ocupaciona,Ocupacionb,Ocupacionc, TasaRecoleccion
execute 'gdxxrw.exe  results.gdx o=results.xls var=VPN.l rng=Hoja1!B5
